Modelling under EViews.
In this second part, we will develop the way the previous tasks can be produced using the EViews software.
This is the method we shall use.
We shall consider each task in turn:
· Accessing the data.
· Transforming it into the model concepts.
· Producing the model framework.
· Running simulations over the past.
· Producing a “forecast”.
· Testing the consequences of shocks on the assumptions.
To each task is assigned an EViews program. These programs are generally too long, and their statements too repetitive, to be presented in detail. Instead we shall :
· Give a general methodology for each task.
· Identify elements which deserve to be developed. In case of repetitions (such as data retropolations on equation estimations) we shall only present them once.
Reading the data : read_afs.prg
The very first task in the modelling process is the transfer of the original data into the modelling software : EViews. For this the best organization in our opinion is the following :
· Access the original file (normally in Excel format).
· Copy it under another name.
· Create a new sheet, and copy the original data into it (or link the cells to the original data).
· Insert series names in the line (or column if series are organized in columns) closest to the data matrix. For instance if a sheet with data in columns contains information in the first three columns, and headers in the first two lines, the data will start in cell D3. You should create a new column D with names, moving the first data element to cell E3.
The names need not be the ones used in the model
· Create a new EViews workfile, with a base “page” (EViews equivalent to an Excel sheet) with the same periods and periodicity as the Excel sheet (or the largest sample considered if you read from several sheets with different starting and ending dates.
· Copy the data into this page.
· If necessary, repeat the process for all the input sheets.
Application to our model
To build the South African model, we shall use mostly data from the World Bank (as explained in the first document).
The original file contains 1300 series per country, and 152 countries or groups. The series are organized by line.
We have first selected the 1300 series associated with South Africa (to avoid manipulating too large a file) and stored with the name “afsud_wb.xls”.[footnoteRef:1] [1:  We shall use the Excel 1997-2003 format, to make the program compatible with older versions of EViews.] 

· We create a workfile through
wfcreate(wf=data_afs,page=wb,t) a 1960 2020    
where
data_afs is the name of the workfile.
Wb	 is the name of the first page
t 	means that series are in columns (default : in lines)
a 	means that the data is annual.
1960 and 2020 are the starting and ending dates.
Note that the World Bank data actually ends in 2012. We just want to be allowed to produce computations after that date.
· We read the data through

read(t=xls,e2,s=data,t) afsud_wb.xls 1300
where 
t=xls 	states the file type
e2	is the upper left cell in the data matrix
s=data	states the name of the Excel sheet
t 	means that data is in columns
afsud_wb.xls  is the name of the file
1300 	is the number of series
Note that the period read is commanded by the last SMPL statement: 
SMPL 1960 2012
The data in cell E2 will be associated to 1960 by EViews, independently from any line of dates.
Our task is simplified by the fact that names appear on the column left of the first data. Even if they are a little esoteric, we shall keep them (our model will use more standard ones).
· We create a group for all the series in the page:
group g_wb *
except for the generic RESID series.
We shall not describe the statements further, except for the last one which save the workfile under the name
data_afs.wf1
Note : we delete the page “ilo” to start from a blank one.
The full program

'        This program transfers the  data for South Africa to EViews
'        The data comes from the World bank and the International Labor Organization

'        We define the directory

cd "c:\users\jean louis brillet\documents\eviews\book_africa\afsud"

'        We close the file in case it exists

close data_afs    

'        We create a data file for the World Bank series (1)

wfcreate(wf=data_afs,page=wb,t) a 1960 2020    

'         We delete all preexisting series

delete *

'         We read South African data from an Excel file called "afsud_wb" with a page named "data"

smpl 1960 2013
read(t=xls,e2,s=data,t) afsud_wb.xls 1300

'         We read World data from an Excel file called "world_wb" with a page named "wb" 

smpl 1960 2013
read(t=xls,e2,s=data,t) world_wb.xls 3

'        We create a group for the original series

group g_wb *

'        We drop the "resid" series

g_wb.drop resid

'       We save the first version of the workfile

wfsave data_afs

'       Now we treat the ILO data

pagedelete ilo

'      It will run from 1974 to 2012

pagecreate(page=ilo) a 1974 2012

'      The format is very special
'      Series are organized by year
'      We have only one observation by line
'      and the number of series varies with each year

'      Example

'      From 1974 to 1979 we have 8 series

'    We start counting the lines

!j=1
for !i=1974 to 1979

'     We establish the sample

smpl !i !i 

'      We read the 8 values in column E

read(type=excel,t,e{!j},,s=data) w_afs.xls 8

'      We increment the line counter by 8
 
!j=!j+8
next    

'      We move to the next set of periods

for !i=1980 to 1992
smpl !i !i 
read(e{!j}, type=excel,t,s=data) w_afs.xls 13 
!j=!j+13
next 

for !i=1993 to 1993
smpl !i !i 
read(e{!j}, type=excel,t,s=data) w_afs.xls 9 
!j=!j+9
next 

for !i=1994 to 2002
smpl !i !i 
read(e{!j}, type=excel,t,s=data) w_afs.xls 16 
!j=!j+16
next

for !i=2005 to 2005
smpl !i !i 
read(e{!j}, type=excel,t,s=data) w_afs.xls 16 
!j=!j+16
next

for !i=2006 to 2006
smpl !i !i 
read(e{!j}, type=excel,t,s=data) w_afs.xls 15 
!j=!j+15
next

for !i=2007 to 2008
smpl !i !i 
read(e{!j}, type=excel,t,s=data) w_afs.xls 16 
!j=!j+16
next

for !i=2009 to 2010
smpl !i !i 
read(e{!j}, type=excel,t,s=data) w_afs.xls 19 
!j=!j+19
next

for !i=2011 to 2012
smpl !i !i 
read(e{!j}, type=excel,t,s=data) w_afs.xls 9 
!j=!j+9
next

'    We define a group for all series
'     except for resid

group g_ilo *
g_ilo.drop resid

'     We save the file again

wfsave data_afs 
generating the data: genr_afs.prg
Our next task will be to generate the series needed by our model.
For this we need to access the elements obtained in the previous phase, and apply in sequence the statements allowing to generate a full set of model series.
This time, the program is too long and too repetitive to describe all its elements. We shall only present relevant issues once.
In this program, we shall :
· Create a page associated with the model.
· Define the original variables as linked with the original pages.
· Create the series needed by the model, through transformations of the original series.
In the course of this process, we shall have to :
· Retroplate, interpolate and extrapolate series.
· Define scalars for the base period.
application to our model
First, we have to create a page with the relevant characteristics, and make sur it contains no element. 
pagecreate(page=model) a 1960 2020
delete *
· Then we copy the list of variables in each page (created in the previous program) to the model page. For instance for the page wb with the group g_wb, we use:
copy(g=l) wb\g_wb  
“g=l” means that we copy only the group definition, not the series. No series wil appear yet in the page “model”.
· We create elements in the page linked to the original series, with the same name except for the prefix “v_“. This will allow to separate the original and model series. We will use for the model variables the prefix “a_” in case we have to make comparisons with elements from other countries.[footnoteRef:2] [2:  In the model text we will drop the prefix (this is quite easy).] 

“link” creates the variables as “linked”. They exist only as a reference to another one (and will follow automatically its changes).
“linkto” defines the link itself, for instance between v_ag_agr_trac_no to and wb\ag_agr_trac_no to 

smpl 1960 2020.
for !i=1 to g_wb.@count
%1=g_wb.@seriesname(!i)
link v_{%1}
v_{%1}.linkto wb\{%1}
next   
· We start creating the elements, through GENR. We will not describe the sstatements in detail. Let us only say that :
· As the program develops, more and more statements will call for model variables.
· We create a time trend, with the value of the year. This will be quite useful later.
· We shall have also to produce scalars, generally as base year values. This is done by :
@elem(name, “period”) such as the deflator of foreign currency:
genr a_er=a_ern/@elem(a_ern,"2005")
·  In some cases, we have to compute missing elements through n-polation. For this, we need to create scalars. For instance, the number of wage earners is not known in 2005 and 2006. We will interpolate it using the share in total employment in 2004 and 2007.
scalar a_rls2=@elem(a_ls,"2004")/@elem(a_lt,"2004")
scalar a_rls3=@elem(a_ls,"2007")/@elem(a_lt,"2007")
smpl 2005 2006
genr a_ls=a_lt*(a_rls2+(a_t-2004)/3*(a_rls3-a_rls2))
· In our case, we need assumptions on some unknown elements :
· Housing investment.
· Social constributions by workers
· Number of civil servants
· The sharing of direct taxes between firms and households.
· Intermediate consumption.
The full program

'       This is an example of data transfer

'      This program will start from the data for South Africa
'       provided by the World Bank

'      The remaining problems will be noted "*****pb"
'      and specific remarks will be noted "*****rem"

'      We decide on the directory
'      This is not generally necessary
'       except if one works on several projects
'      or maintains several directories for the same project
'       It guarantees trial versions do not destroy official ones

cd "c:\users\jean louis brillet\documents\eviews\book_africa\afsud"

'     The original data is contained in a page called "wb" with the original names
'     The page "model" contains links to the "wb" series
'      with the addition of a "v_" prefix

'-------------------------------------------------------------------------
'     This technique can be used with any source 

'      In the best case, if the set available is the same (or larger) than the World Bank set
'      one has just to replace the World Bank names in the following statements
'-------------------------------------------------------------------------

'       we close the original file "data_afs"  
'      in case it is already open 
'      having two versions of the same file open in memory is quite dangenrous...

close data_afs 

'     We open it again

open data_afs

'       We copy the definition of the group to a page called "model"
 
'       We delete the page if it exists

pagedelete model

'       We create the page annual from 1960 to 2020

pagecreate(page=model) a 1960 2020
delete *

'      We copy the definition of the group g_wb (not the series) from the page wb to model
'      This is done by "g=l"

copy(g=l) wb\g_wb   

'       We create linked variables adding the prefix "v_" to the original series names
'       to separate them from the variables we are going to compute

smpl 1960 2020
for !i=1 to g_wb.@count

'     We get the name

%1=g_wb.@seriesname(!i)

'     We  create the linked variable

link v_{%1}

'    We establish the link

v_{%1}.linkto wb\{%1}
next   

'      We copy the definition of the group g_ilo (not the series) from the page wb to model
'      This is done by "g=l"

copy(g=l) ilo\g_ilo   

'       We create linked variables adding the prefix "v_" to the original series names
'       to separate them from the variables we are going to compute

smpl 1960 2020
for !i=1 to g_ilo.@count

'     We get the name

%1=g_ilo.@seriesname(!i)

'     We  create the linked variable

link l_{%1}

'    We establish the link

l_{%1}.linkto ilo\{%1}
next   

'     We generate a time trend

smpl 1960 1960
genr t=1960
smpl 1961 2020
genr t=t(-1)+1
pageselect model

'     We have to make an assumption on the sharing of indirect taxes 
'     into VAT and other indirect taxes
'     as OECD provides only a global variable

'     p_oit = assumption on the share of oit in indirect taxes

'     indirect taxes are not separated into VAT and other taxes

'    Our series will have the prefix "a_" for South Africa
'    this is to separate them from series for other countries if they share the same page
'    The prefix can be easily discarded later
'
scalar a_p_oit=0.2

scalar txq=0.05
scalar txn=0.01
scalar txp=0.05

'     we create a time trend
'     with the value of the year for the first quarter
'     to which we add 1 for each following quarter of the year 

'   This will be quite useful to :

'      create yearly time trends
'      replace actual dummy variables by expressions using logical conditions
'      much easier to manage

smpl 1960 1960
genr a_t=1960
smpl 1961 2020
genr a_t=a_t(-1)+1

'     Now we start with the supply - demand equilibrium

smpl 1960 2020
genr a_usd=v_PA_NUS_FCRF
genr a_gdpv=v_ny_gdp_mktp_cn
genr a_gdp=v_ny_gdp_mktp_kn

'   The model separates market GDP


genr a_cohv=v_ne_con_petc_cn
genr a_coh=v_ne_con_petc_kn
genr a_pcoh=a_cohv/a_coh

genr a_cgw=0
genr a_gdpmv=a_gdpv-a_cgw
genr a_gdpm=a_GDP-a_CGw/a_pcoh
genr a_pgdpm=a_gdpmv/a_gdpm

'      Trade af current and constant prices, associated deflators

genr a_mv=v_ne_imp_gnfs_cn
genr a_m=v_ne_imp_gnfs_kn
genr a_pm=a_mv/a_m
genr a_xv=v_ne_exp_gnfs_cn
genr a_x=v_ne_exp_gnfs_kn
genr a_px=a_xv/a_x

'      Final demand, VAT and value added (excluding VAT)
'      This separation is important as VAT will apply to final demand

genr a_fd=a_GDPm+a_M-a_X
genr a_fdv=a_GDPmV+a_MV-a_XV
genr a_pfd=a_fdv/a_fd
genr a_oit=(1-a_p_oit)*v_ny_tax_nind_cn 
genr a_vat =a_p_oit*v_ny_tax_nind_cn 
genr a_r_vat =a_vat/(a_fd*a_pfd-a_vat)

'       value added (excluding VAT)

genr a_qv=a_gdpmv-a_vat
genr a_r_oit=a_oit/(a_qv-a_oit)
scalar a_r_vat0=@elem(a_r_vat,"2005")
genr a_q = a_gdpm-a_r_vat0*a_fd/(1+a_r_vat0)
genr a_pq=a_qv/a_q

'      Investment

'    *****pb : government investment

genr a_ihv=0      '    *****pb : no measure of housing investment
genr a_ih=0        '    *****pb : no measure of housing investment
genr a_ipv=v_ne_gdi_fprv_cn
genr a_igv=v_ne_gdi_ftot_cn- v_ne_gdi_fprv_cn 
genr a_ip=v_ne_gdi_ftot_kn*a_ipv/(a_ipv+a_igv+a_ihv)
genr a_ig=v_ne_gdi_ftot_kn*a_igv/(a_ipv+a_igv+a_ihv)
genr a_pip=a_ipv/a_ip
genr a_pig=a_igv/a_ig
genr a_dep=v_ny_adj_dkap_cd   ' *****rem additional measure of depreciation in USD

'    Capital and capacities

genr a_cap=a_q     ' *****rem : will be computed after estimation of production function
smpl 1964 2013
genr a_k=2*a_q
genr a_rdep = 0.05
smpl 1965 2013
genr a_k=a_k(-1)*(1-a_rdep)+a_ip
smpl @all

'     The rate of use

genr a_ur=a_q/a_k'-1) 
genr a_pk=a_cap/a_k(-1)
genr a_urd=1

'      We assume that for each unit of value added we need one unit of intermediary consumption 
'      this is generally not too far from the truth

genr a_tc=1
genr a_ci=a_tc*a_q

'      Total demand

genr a_td=a_fd+a_ci

'    Elements of demand 

'       Changes in inventories

genr a_icv=v_ne_gdi_stkb_cn
genr a_ic=v_ne_gdi_stkb_kn
genr a_pic=a_icv/a_ic

'        Government consumption (includes wages of civl servants)

genr a_cogv=v_ne_con_govt_cn
genr a_cog=v_ne_con_govt_kn
genr a_pcog=a_cogv/a_cog
genr a_r_pcog=a_pcog/a_pfd

'         Final demand of Government

genr a_fdgv=a_cogv+a_igv
genr a_fdg=a_cog+a_ig
genr a_gd=a_ig+a_cog
genr a_fdxr=0    '    *****rem : look further into the composition of demand

'    employment, unemployment, population

genr a_popt=v_sp_pop_totl
genr a_pop1564=a_popt*v_sp_pop_1564_to_zs/100
genr a_lt=v_sl_emp_totl_sp_zs*a_pop1564/100
genr a_lg=0.10*lt         '  *****pb : no measure of civil servants. Will have to be changed
genr a_lf=a_lt-a_lg
genr a_pl=a_q / a_lf
genr a_un=v_sl_uem_totl_zs*v_sl_tlf_totl_in/100    '   *****rem : un known only in 2006
genr a_unr=a_UN/(a_lt+a_un)*100  
genr a_popac=a_lt+a_un
genr a_lp=a_q/a_lf

'    wages and further deflators

genr a_ls=v_sl_emp_self_zs*a_lt/100  

'    We interpolate the number of self employed

'     First we compute the share as scalars for the years for which we have the information

scalar a_rls1=@elem(a_ls,"2003")/@elem(a_lt,"2003")
scalar a_rls2=@elem(a_ls,"2004")/@elem(a_lt,"2004")
scalar a_rls3=@elem(a_ls,"2007")/@elem(a_lt,"2007")
scalar a_rls4=@elem(a_ls,"2011")/@elem(a_lt,"2011")

'    We use the first scalar to retropolate 

smpl 1991 2002
genr a_ls=a_lt*a_rls1

'    The second and the third to interpolate 

smpl 2005 2006
genr a_ls=a_lt*(a_rls2+(a_t-2004)/3*(a_rls3-a_rls2))

'    The last to extrapolate

smpl 2012 2012
genr a_ls=a_lt*a_rls4

'     We compute the number of wage earners

smpl @all
genr a_lw=a_lf-a_ls
genr a_r_lw=a_lw/a_lf 

'     The production price is the deflator of production 
'     using a demand deflator excluding vat


genr a_pfdxt=a_pfd*(1+a_r_vat0)/(1+a_r_vat)
genr a_tc=1     ' ******rem : assumption
genr a_pp=(a_pq+a_tc*a_pfdxt)/(1+a_tc) 

genr a_r_scf =0      ' ****** unknown social contributions. Will have to be changed.
genr a_scf=0

'      The wage rate

'     It is not known in 2003 and 2004
'     we interpolate it

genr a_wr_man=l_w_ima
smpl 2003 2004
genr log(a_wr_man)=@elem(log(a_wr_man),"2002")+(t-2002)/3*(@elem(log(a_wr_man),"2005")-@elem(log(a_wr_man),"2002")) 

'      After 2008 we make it grow like the consumption price index

smpl 2009 2012  
genr @pch(a_wr_man)=@pch(a_pcoh) 
smpl @all 

'   The yearly value

genr a_wr=a_wr_man*12

'    Total wages

genr a_wage=a_wr*(a_lt-a_ls)
genr a_r_scw=0.10        '  *****pb : no measure of contributions by employees '
genr a_scw=a_r_scw*a_wage 
genr a_wagef=a_wr*(a_lf-a_ls)
genr a_wageg=a_wr*a_lg
genr a_uwc=a_wr*(1+a_r_scf)/a_lp

'      The demand price excluding VAT

genr a_ern=v_pa_nus_fcrf 

'       The exchange rate as a deflator

genr a_er=a_ern/@elem(a_ern,"2005")
genr a_erx=a_er

'       ratios between demand prices

genr a_r_pip=a_pip/a_pfd
genr a_r_pcoh=a_pcoh/a_pfd
genr a_r_pig=a_pig/a_pfd

'    interest rates
 
genr a_irs=v_fr_inr_dpst    
genr a_irl=v_fr_inr_dpst     
genr a_ir=0.5*(a_irs+a_irl)
genr a_irm=a_ir 
genr a_irsx=a_irs
genr a_irx=a_ir
genr a_irr=a_ir-100*@pch(a_pcoh)
genr a_irl_ec=0
genr a_irsr=a_irs-100*@pchy(a_pcoh)
genr a_irst=0 

'    The relative cost

genr a_relc=a_wr*(1+a_r_scf)/a_pip/(a_ir/100-@pchy(a_pcoh)+a_rdep)
smpl 1978 2013
genr a_irm=0.2*a_ir+0.8*a_irm(-1) 
genr a_irmx=a_irm

'---------------------------------------
'     Households
'---------------------------------------

smpl @all
genr a_rem=v_BX_TRF_PWKR_DT_GD_ZS*a_gdpv/100

'      Remittances are extrapolated in 2011 and 2012

smpl 2011 2012
genr a_rem=a_rem(-1)*a_gdpv/a_gdpv(-1)
smpl @all

'      remittances

genr a_remx=a_rem/a_er
genr a_socb=v_sh_xpd_publ_zs*a_gdpv/100    '  *****rem : govt health expenditures
genr a_socbr =a_socb/a_pcoh/a_popt

'      Social benefits are retropolated iuntil 1994

smpl @first 1994
genr a_socb =@elem(a_socb,"1995")*a_gdpv/@elem(a_gdpv,"1995")
genr a_socbr =a_socb/a_pcoh/a_popt
smpl @all

'      Direct taxes are shared between firms and households

genr a_ifp=0.5*v_gc_tax_ypkg_cn   ' ****rem : division 0.5 
genr a_ict=0.5*v_gc_tax_ypkg_cn
genr a_subs=v_gc_xpn_trft_cn  '   *****rem :  
genr a_tar=v_gc_tax_impt_cn   '   ***** rem  
genr a_tar=v_gc_tax_intt_cn  '   *****rem : which is the right series?

'      Rates

genr a_r_subs = a_subs/a_qv
genr a_r_tar=a_tar/a_mv  

'      Retro and extra polations

smpl @first 1999
genr a_r_subs=@elem(a_r_subs,"2000")
smpl 2012 2012
genr a_r_subs=@elem(a_r_subs,"2011")
smpl 2012 2012
genr a_r_tar=a_r_tar(-1)
genr a_socbr=a_socbr(-1)*(1+txq)/(1+txn)
genr a_socb=a_socbr*a_popt*a_pcoh

'     We recompute values over the full period

smpl @all
genr a_subs=a_r_subs*a_qv 
genr a_tar=a_r_tar*a_mv

'     Th margins

genr a_marg=(a_qv*(1-a_r_oit+a_r_subs)-a_wagef*(1+a_r_scf))

'      Assumptions on household revenue linked to production (revq) and not (revx)
 
genr a_revx=0.1*a_gdpv    '  *****pb : no measure of non wage revenue
genr a_revq=0.7*a_marg     '  *****pb : no measure of non wage revenue

'      Household total revenue

genr a_hi =a_wage-a_scw+a_socb+a_revx+a_revq+a_rem 

'     The income tax rate

genr a_r_ict=a_ict/a_hi    '    *****pb : only global tax on revenue 

'      Retro and extra polations

smpl 2012 2012
genr a_r_ih=a_r_ih(-1)
genr a_r_ict=a_r_ict(-1)
smpl @first 1999
genr a_r_ict=@elem(a_r_ict,"2005")
genr a_r_tar=@elem(a_r_tar,"2005")

'      We recompute the values

smpl @all
genr a_ict=a_r_ict*a_hi
genr a_tar=a_r_tar*a_mv
genr a_hdi=a_hi-a_ict
genr a_hrdi=a_hdi/a_pcoh
genr a_sr=1-a_coh/(a_hrdi)
genr a_r_revx=a_revx/a_pfd
genr a_r_revq=a_revq/a_qv
genr a_wageg=a_wr*a_lg       '   ****rem : l_g not available 
genr a_r_ih=a_ih/a_hrdi

'    rates

genr a_r_scg=  a_r_scf
scalar a_r_oit0=@elem(a_r_oit,"2005")

'----------------------------------------------------
'     external trade
'----------------------------------------------------

genr a_r_tarx=0
scalar a_r_tarx0=@elem(a_r_tarx,"2005")
scalar a_r_tar0=@elem(a_r_tar,"2005")

'     Foreign price and foreign demand

genr a_ppx=v_NY_GDP_MKTP_CD/v_NY_GDP_MKTP_KD   '    rem***** look for better later 
genr a_wd =v_NE_IMP_GNFS_KD   '    rem***** look for better later  

'      Trade ratios

genr a_rcvol=a_x/a_m
genr a_rcval=a_xv/a_mv
genr a_ttrad=a_px/a_pm
genr a_trb=a_xv-a_mv

'       Interests

genr a_nixs=v_dt_int_dstc_cd*a_ern
genr a_nixl=v_dt_int_dlxf_cd*a_ern  
genr a_nix=a_nixs+a_nixl
genr a_nixd=0.5*a_nix   '  *****rem : separating local and foreign currencies
genr a_nixx=0.5*a_nix   '  *****rem : separating local and foreign currencies
genr a_NIXD_ec=(a_NIXD-(a_NIXD(-1)*a_IRM/a_IRM(-1)-a_IR/100*0.5*a_TRB))/(a_xv+a_mv)
genr a_NIXX_ec=(a_NIXX-(a_NIXX(-1)*a_IRMX/a_IRMX(-1)*(a_er/a_er(-1))-a_IRX/100*0.5*a_TRB))/(a_xv+a_mv)
genr a_fcapx=a_trb-a_nix

'      Price and competitiveness

genr a_PMT=a_PM*(1+a_r_tar)/(1+a_r_tar0)
genr a_compm=a_pmt/a_pp
genr a_COMPX=a_PX*(1+a_r_tarx)/(1+a_r_tarx0)/(a_PPX*a_ER)

'--------------------------------------------------
'      firms account
'--------------------------------------------------

genr a_rmarg =a_marg / a_qv
genr a_rprob = a_marg/(a_pfd*a_k(-1)) 
genr a_r_ifp=a_ifp/(a_marg-a_revq)  

'    We retropolate the tax on profits
  
smpl @first 1999
genr a_r_ifp=@elem(a_r_ifp,"2000")
smpl 2012 2012
genr a_r_ifp=a_r_ifp(-1)
smpl @all
genr a_ifp=a_r_ifp*(a_marg-a_revq)

'     The interests paid

smpl 1991 1991
genr a_nif=(-a_marg+a_revq+a_ifp)*0.05
smpl 1992 2013
genr a_fcapf1=a_marg-a_revq-a_ifp-a_pip*a_ip-a_pfd*a_ic
genr a_nif=a_nif(-1)*a_irm/a_irm(-1)-a_fcapf1

'      Balances

smpl 1960 2013
genr a_prof=a_marg-a_revq-a_ifp-a_nif
genr a_rprof=a_prof/(a_pfd*a_k(-1)) 
genr a_fcapf=a_prof-a_pip*a_ip-a_pfd*a_ic
genr a_i=a_ip+a_ih+a_ig
genr a_fdxr=(a_fd-(a_COH+a_IP+a_IH+a_IC+a_cog+a_IG))/a_q 
genr a_fcapf=a_prof-a_pip*a_i-a_pfd*a_ic 
genr a_nif_er=(a_nif-(a_NIF(-1)*a_IRM/a_IRM(-1)-a_IR/100*a_FCAPF))/a_qv 

'------------------------------------------
'     Budget elements
'------------------------------------------

[bookmark: _GoBack]genr a_fcapg=-v_fs_ast_cgov_gd_zs*a_gdpv/100  
genr a_r_oit=a_oit/(a_qv-a_oit)
genr a_nig=v_gc_xpn_intp_cn/100

'    The intersts paid

smpl 1991 1991
genr a_nig=0
smpl 1992 @last
genr a_nig=a_nig(-1)*a_irm/a_irm(-1)-a_ir/100*a_fcapg
smpl @all

'      Various elements

'    The wage rate is the same as firms'

genr a_wg=a_wr

'      Social contributions

genr a_scg=a_wg*a_r_scf


genr a_revg=a_ict+a_oit+a_vat+a_scf+a_scg+a_tar+a_scw+a_ifp
genr a_expg=a_revg-a_fcapg
genr a_dfgv=a_igv
genr a_r_expg=(a_EXPG-(a_FDGV+a_WAGEG+a_SUBS+a_SOCB+a_NIG+a_Scg))/a_Qv 

'      Additional potential residuals

genr a_r_revg=0
genr a_r_recg=0

'     Balances

genr a_expg=a_revg-a_fcapg
genr a_fcapgp=100*a_fcapg/a_gdpmv
genr a_nig_er=(a_nig-(a_nig(-1)*a_irm/a_irm(-1)-a_ir/100*a_fcapg))/a_qv 

'      These elements will come from estimations
'       Defining them will allow to create the groups of model variables

genr a_cost=na
genr a_costw=na
genr a_lpt=na

wfsave data_afs
creating the model : 
Now that we have created the data, we can specify the model.
Of course, this sequence does not represent a realistic case, in which you will need a lot of iterations between data gathering, data transformations, model specification, equation estimation, and model testing through simulations and shocks.
We shall work in two phases.
First we will define a logical model framework, in which we  completely specify the identities but define only the influences in the behavioral equations. This will allow us to produce already a series of tests, checking for errors before we proceed further.
Then we will replace the logical behaviors by actually estimated equations.
Creating the model framework: 
This program:
Removes the “a_” suffix from the model series names.
· Creates a blank model called _mod_afs (the underscore guarantees that the model icon can be accessed at the beginning of the page).
· Introduces the identities using:

   _mod_afs.append text-of-the equation

Introduces the behavioral equations as:

   _mod_afs.append dependent=f*(sum of the explaining elements).

For instance we will have :

    _mod_afs.append cap=f*(k+lf+t)

This is as close as possible to the formula we would prefer:
cap=f (k,lf,t)

while conforming to the EViews syntax.

f is defined as a scalar to avoid being considered as exogenous.
Using the results
Even if the model is far from complete, we can already perform tests which will avoid following the wrong path, which means losing time in the best of cases, but also ending with a wrong model.
· We can check that identies are consistent with the data using:
_mod_afs.append assign @all _c
_mod_afs.solve(d=f) 
for !i=1 to _g_vendog.@count
%1=_g_vendog.@seriesname(!i)
genr dc_{%1}={%1}-{%1}_c
genr pc_{%1}=100*dc_{%1}/({%1}+({%1}=0))
next

“d=f” characterizes the method
“_c” is the suffix

The loop computes level and percentage differences (must be zero).
The use of the Boolean expression « +({%1}=0) », giving 1 if true and 0 if false, ensures that for null series (for instance a potential tax) the division of 0 by zero gives 0 as 0/(0+1).
If the variable is not null the computation will not be affected as the expression will give 0 (false).
This trick is quite useful in many situations.
· We can check the list of variables by clicking on the model icon, and selecting “View” then “Variables”. In particular, we can identify the presence of parasites (a typing mistake?) or the exogeneity of logically endogenous elements.
· We can check the block structure of the model by clicking on the model icon, and selecting “View” then “Block structure”. If our ideas succeed, this structure will apply to the future model, so any problem identified can be corrected now, before any estimation. For instance, one can detect the absence of an expected loop, or the fact that a variable has no retroaction.
In our case, we get :
umber of equations: 95
Number of independent blocks: 6
Number of simultaneous blocks: 3
Number of recursive blocks: 3
Largest simultaneous block: 54 Equations (4 feedback vars)
Block 1: 54 Simultaneous Equations (4 feedback vars)
 	pcoh(24)                 	k(10)                    	lf(5)                    	cap(3)                   	ur(4)                    	irs(32)                  	lt(7)                    	popac(12)                	un(13)                   	unr(14)                  	lp(8)                    	uwc(15)                  	pq(16)                   	qv(49)                   	revq(58)                 	lw(6)                    	wagef(52)                	wageg(40)                	wage(41)                 	scw(42)                  	socb(39)                 	revx(38)                 	er(31)                   	rem(37)                  	hi(43)                   	ict(44)                  	hdi(45)                  	hrdi(46)                 	coh(48)                  	pfdxt(23)                	pp(17)                   	pm(29)                   	ci(67)                   	ih(47)                   	ic(11)                   	fd(66)                   	td(68)                   	pmt(64)                  	compm(65)                	m(69)                    	mv(72)                   	px(30)                   	compx(70)                	x(71)                    	gdpm(1)                  	fdl(20)                  	xv(73)                   	vat(82)                  	gdpmv(50)                	fdlv(19)                 	ip(9)                    	wr(27)                   	q(2)                     	pfd(22)                 
Block 2: 16 Recursive Equations
 	pip(25)                  	pig(26)                  	pcog(89)                 	fdv(18)                  	pfdl(21)                 	cost(28)                 	irl(33)                  	ir(34)                   	irm(35)                  	relc(36)                 	pgdpm(51)                	subs(53)                 	marg(54)                 	rmarg(55)                	ifp(56)                  	i(63)                   
Block 3: 3 Simultaneous Equations (1 feedback var)
 	prof(59)                 	fcapf(62)                	nif(57)                 
Block 4: 18 Recursive Equations
 	rprof(60)                	rprob(61)                	rcval(74)                	rcvol(75)                	ttrad(76)                	trb(77)                  	nixd(78)                 	nixx(79)                 	nix(80)                  	fcapx(81)                	scf(83)                  	oit(84)                  	tar(85)                  	scg(86)                  	revg(87)                 	igv(88)                  	cogv(90)                 	fdgv(91)                
Block 5: 3 Simultaneous Equations (1 feedback var)
 	expg(93)                 	fcapg(94)                	nig(92)                 
Block 6: 1 Recursive Equations
 	fcapgp(95)              

In the above results, we detect the presence of a large loop, but also that it excludes 41 elements, which can be computed later (with small loops for firms and government between expenditures, balance and interests paid).
The crucial point here is to check that this exclusion is logical. Indeed it is, if we look closely at the definition of the GDP and final demand deflators, the global value of final demand, and the profitability elements. A more sophisticated framework (like a role of profitability in investment) could bring back some of these elements into the loop. 
You are welcome to make this test by changing the list of explanatory elements in behavior definition.
Estimating the equations and storing the result
We have now produced a model in which:
· The economic framework looks logical.
· The identities are consistent with the data.
· This data also allows to estimate the equations.
This is the last task we need to perform before we have produced a full version.
Application to our model
In this example, we shall limit ourselves to simple methods, considering that for the countries we consider the samples are limited and the economy has not been so stable in the past.
But even in this case, we propose two approaches, associated with the search for an acceptable formula by trial and arror, and the introduction of the elected element in the model.
Looking for the best formula.
This is the only case in which we do not suggest to use a program, especially if we are testing a single equation. But not menus either….
What we propose is to use the command line, to specify the equation, using explicit coefficients in case of non-linear expressions[footnoteRef:3] [3:  Relative to coefficients. The equation can use series expressions as complex as necessary, if it is linear in coefficients the method is still ordinary least squares.] 

If you are using least squares (ordinary or not) you will just state “ls” before the formula. For an import equation using total demand TD and price competitiveness COMPM as explanatory variables, we can use:
ls log(m) log(td) c
or
coef(1) c_m
ls log(m)=c_m(1)*log(td)+c_m(2)
or even
ls dlog(m)=c_m(1)*dlog(td)+c_m(2)*dlog(compm)+c_m(3)*(log(m(-1)/td(-1))-c_m(4)*log(compm(-1))+c_m(5)
This approach has two good points:
· You can edit the statement and run it again (copy it before if you do not want to lose the initial text).
· The command window will keep all you statements. You can visualize them and copy them as a program block when the time is right (for instance you are stopping for lunch).
Storing the accepted formula
When you have reached a acceptable formula (and you do not want to look further), it is necessary to formalize the process. We propose the following:
· Define the name of the dependent variable as the root element. 
· Create a vector of coefficients adding the prefix “c_” to the root.
· Create a vector of parameters adding the prefix “p_” to the root.
· Generate a residual by adding “e_” to the root.
· Create the equation by adding “_eq_” to the root, and specifying the text.
If the equation should be estimated immediately, you can add “.ls” to the name, and (p) if you want the reults to be “printed” in an external file.
· Store the estimated residual.
For instance for the above example, we can use :

  coef(10) c_m
  vector(1) p_m
  genr m_ec=0
  equation _eq_m.ls(p) dlog(m)=c_m(1)*dlog(td)+c_m(2)*dlog(compm)+c_m(3)*(log(m(-1)/td(-1))-c_m(4)*log(compm(-1))+c_m(5)+m_ec
  genr m_ec=resid

This method ensures in particular that :
· The left hand and right had sides should give the same result. This means the residual check” presented earlier can now apply to all equations.[footnoteRef:4] [4:  The sources of error are more limited here. One can thnk essentially of forgetting to resestmate equations following changes in the data values.] 

· Switching between estimations and calibrations is made easy: one just has to replace “c” by “p” and vice versa (and specifying the value of course).
Other methods
Although they are not used here, three alternate techniques can be considered: cointegration, system estimation, and panel estimation. They are described in detail in my book availabla at:
http://www.eviews.com/StructModel/structmodel.html
solving the model over the past
Once the model is complete, it can be simulated over the past. As explained before, the main purpose of this task is to check that the results (using the most probable residual: 0) are not too far from historical values. This test allows only to discard wrong models, but to validate any.
One can also check immediately the sensitivity to shocks, before any forecast. The idea is always to locate problems as soon as possible, to avoid treading the wrong path.
We give here a program which allows to compute the average error, replacing the estimated residuals either by zero (long term solutions) or by the residuals of the previous period (short term).
As to shocks, we shall present them later, with the simulations over the future.
The program

'     This program tests the precision of the model
'      using simulations over the past with unknown residuals

'      First case : long term simulations with no information, the residual is set to zero (the most probable value).
'      Second case : short term simulations knowing the residual from the previous period: it is use in the presnt one.

'      We open the model file (once)

close model_afs
open model_afs

'     We select the model page
'      
pageselect model

'     We will consider 10 periods

smpl 2003 2012
_mod_afs.scenario "scenario 1"

'     g_res0 is the group of all residuals

group g_res0 COH_EC IC_EC IP_EC IRL_EC LF_EC M_EC NIXD_EC NIXX_EC PM_EC POPAC_EC PQ_EC PX_EC WR_EC X_EC 

'     g_res is the group of residuals considered currently
'     It can be edited at will, as one can always go back to g_res0

delete(noerr) g_res 
group g_res COH_EC IC_EC IP_EC  LF_EC POPAC_EC M_EC X_EC  PQ_EC PX_EC  PM_EC WR_EC
 
smpl 2003 2012

'    We create a blank list of overriden assumptions

_mod_afs.override  

'     We create  a loop on residuals

for !i=1 to g_res.@count
%1=g_res.@seriesname(!i)

'     We create zero residuals with the suffix "_y"
'     We create last period residuals with the suffix "_z"

genr {%1}_y=0
genr {%1}_z={%1}(-1)

'     We add each residual to the list

_mod_afs.override(m){%1}  
next  

'      We simulate the model with the suffix "_y"

_mod_afs.append assign @all _y
_mod_afs.solve(d=d,n=t,i=p,o=n)

'      We simulate the model with the suffix "_z"

_mod_afs.append assign @all _z
_mod_afs.solve(d=d,n=t,i=p,o=n)

'      We create the difference between simulated and historical
'      in absolute and relative values
'      for all the endogenous variables

 for !i=1 to _g_vendog.@count
%1=_g_vendog.@seriesname(!i)
genr drz_{%1}={%1}-{%1}_z
genr prz_{%1}=100*drz_{%1}/({%1}+({%1}=0))  
genr dry_{%1}={%1}-{%1}_y
genr pry_{%1}=100*dry_{%1}/({%1}+({%1}=0))  
next 

'       We concentrate on 7 variables

'      The gross domestic product
'       Imports
'       Exports
'       Final demand
'       Household consumption
'       Productive investment
'       The household consumption deflator

'       Other elements can be added if needed.

'        We create a 7x2 matrix

matrix(7,2) _tpr

'       We create the group of tested variables

group g_pr gdpm m x fd  coh i pcoh

'       We fill the matrix with the values of the absolute means of the percentage errors.

smpl 2003 2012
 for !i=1 to g_pr.@count
%1=g_pr.@seriesname(!i) 
scalar mprz_{%1}=@mean(@abs(prz_{%1}))
scalar mpry_{%1}=@mean(@abs(pry_{%1}))
_tpr(!i,1)=mprz_{%1}
_tpr(!i,2)=mpry_{%1}
next

'      We create a table called "_pr"  from the matrix "_tpr"

delete(noerr) _pr
freeze(_pr) _tpr

'    The table has three columns
'     The first one contains the variable names
'     The first three lines are not useful

_pr.deleterow(3)

'     We replace the first column with variable titles

_pr(3,1)="GDP"
_pr(4,1)="Imports"
_pr(5,1)="Exports"
_pr(6,1)="Final demand"
_pr(7,1)="Household consumption"
_pr(8,1)="Investment" 
_pr(9,1)="CPI" 

'     And the first line with column definitions

_pr(1,2)="Error 0"
_pr(1,3)="Error lag"

_pr.setjust(A1:A14) left 

'     We display the table

show _pr

'     We set the sample to the period simulated (for subsequent displays).

 smpl 2003 2012

producing forecasts and shocks
One could suppose the following sequence;
· Producing a first simulation to check that the model gives solutions in the short and long run.
· Improving this solution to make it an acceptable forecast.
· Introducing changes in the assumptions to assess their consequences on the economic equilibrium.
Unfortunately, this sequence will fail as the shocks are sure to evidence economic problems, leading to changes in the formulations. Of course, the new forecast using the previous assumptions will no longer be acceptable.
So the last two stages in the sequence will have to be inverted:
· Introducing changes in the assumptions to assess their consequences on the economic equilibrium of the basic forecast.
· Once the properties are acceptable, improving this solution to make it an acceptable forecast.
We can expect that the differences between the two forecasts will not question the tachings of the first shocks (even if they will have to be reproduced of course).
If this is not true, another iteration on the last two steps will be needed, until both the forecast and the shocks are acceptable (a very subjective assessment, unfortunately).
trend simulations : proj_afs0.prg
If the model has been designed correctly, it should converge, for two reasons:
· The error correction format should provide a long term equilibrium, consistent with economic theory.
· If the sort term dynamics follows economic theory too, they should not make the model explode at the beginning of simulations.
To check this in practice, we just need to decide on the assumptions.To check the long term convergence, we need all variables with a common dismension to grow at the same rate. Fortunately, most of the assumptions have no dimension. This concerns:
· The tax rates.
· The residuals in the estimated equations (to avoid heteroscedasticity, most of the estimated expression are either logarithms or ratios).
The only remaining elements are:
· Quantities at constant prices: government demand (a policy instrument) and world growth (a foreign assumption).
· Deflators. This should only represent the foreign prices in foreign currency (if the exchange rate is stable, this should transmit to the local prices.
· The populations. This concerns total population, the potential work force and government employment.
It is clear that if we replace government demand by a given share of GDP, and Govenrment employment by a given share of the potential wok force, all the remaining elements will represent either foreign or structural assumptions
The following gaph show that the solution converges in the long run, to the theoretical growth rates. The convergence speed is very slow howevere, without doubt due to the low values of some error correcting coefficients.



Checking the long term assumptions
When the model reaches a certain size, it becomes difficult to check visually 
· that all assumptions are consistent with their dimension[footnoteRef:5] [5:  For instance if social benefits are considered in purchasing power per head, and the assumption forgets to divide by population.] 

· that for all equations, applying the theoretical growth rates to their elements gives a result growing at the theoretical growth rate[footnoteRef:6]  [6:  For instance the modeler might have forgetten to remove an additional time trend.] 

There is a very simple solution. One just has to generate (as proposed earlier) a full set of exogenous and endogenous elements following the above constraints. Then the model is solved using the “fit” option (separate computation of each equation) and the growth rate of the result is compared to the growth rate of the variable the equation defines. The two must be identical. If not, the elements in the equation are not consistent with its formulation. 
The program we are providing uses this feature.
producing shocks : proj_afs1.prg
The goal of the shocks we will make now is only to check model properties, and to produce a model reliable enough to provide acceptable forecasts. When this is done, we shall repeat the shocks, with results which should not change very much.
Of course, the programs will not change very much too.
In the following program, we will show how to produce several shocks in sequence, each of them applied to a single assumption.
For this, we need to:
· Produce a base simulation.
· Define a list of shocked variables, as a group.
· Define a list of suffixes associated with each shock.
· Create the level of the shocks.
· Simulate the model using the shock.
· Produce tables and graphs.
In our example, the shocks start in 2014. Their level will be:
	letters
	 variable
	definition
	
level

	gd
	ig
	government demand
	+1 point of GDP

	ex
	erx
	the exchange rate
	1 percent devaluation

	vt
	r_vat
	the value added tax 
	-1 point

	tl
	r_tar
	local tariffs
	-1 point

	tx
	r_tarx
	foreign tariffs
	-1 point

	wd
	wd
	foreign quotas
	+1%

	im
	m_ec
	local quotas  
	+1%


The list of letters will be obtained by :

  group shocks_l  ex vt tl tx wd im gd

The list of variables will be obtained by :

  group shocks_v erx r_vat r_tar r_tarx wd m_ec ig

The shocks will be defined by :
 
  genr ig_gd=ig+.01*gdpm_b*(t>=2014)
  genr erx_ex=erx*(1+.01*(t>=2014))
  genr r_vat_vt=r_vat-.01*(t>=2014)
  genr r_tar_tl=r_tar-.01*(t>=2014)
  genr r_tarx_tx=r_tarx-.01*(t>=2014)
  genr wd_wd=wd*(1+.01*(t>=2014))
  genr m_ec_im=m_ec+.01*(t=2014)-.01*(t>2014)*p_m(5)


The last shock deserves an explanation. After 2014, the error correction mechanism extends the shock to the next period, in a proportion p_m(5). To apply only the desired 1% shock, this extension has to be suppressed.
The loop will cover the 7 shocks, producing 6 graphs:

· The supply-demand equilibrium
· The production elements
· Some ratios.
· External trade (2 graphs)
· Prices

And one global table.

Allowing changes in equations
Now it should be clear that the observation of model properties will lead to changes in the equations.
· Because some elements do not show the right properties (for instance if a coefficient should be higher).
· Because some influence is missing, as it could not be estimated correctly; but is essential for model properties (this was explained in the first document).
· Or simply because the modeler wants to know what a given change would bring.
This can be done by creating a new model program, and running it before the new forecast.
However, this process is tedious and error prone. It will be much better to locate the chages in equations in the forecasting program itself.
This will also allow to calibrate some coefficients, either by deciding on a change in influences or even introducing a completely new influence.
To explain this we will use an example, the equation for exports.

  coef(10) c_x
  vector(10) p_x
  smpl 1962 2012
  genr x_ec=0
  smpl @first 2012  
  equation _eq_x.ls(p) dlog(x)=c_x(1)*dlog(wd)+c_x(2)*(log(x(-1)/wd(-1))-c_x(4)*log(compx(-1)))+c_x(4)*dlog(compx)+c_x(5)+c_x(6)*(t-2012)*(t<=2012)+x_ec
  genr x_ec=resid 

We will reproduce the set in  the forecasting program, but leave unchanged all the elements bu the equation itself.
We will replace all the coefficients (named “c_”) but the constant term by “p_”:

  equation _eq_x.ls(p) dlog(x)=p_x(1)*dlog(wd)+p_x(2)*(log(x(-1)/wd(-1))-p_x(4)*log(compx(-1)))+p_x(4)*dlog(compx)+c_x(5)+p_x(6)*(t-2012)*(t<=2012)+x_ec


Now we can introduce changes. For instance we can introduce an impact of the rate of use of capacities, and change the coefficient of competitiveness to -0.7 (a higher value).

  p_x(3)=-0.30 
  p_x(4)= 0.70
  equation _eq_x.ls(p) dlog(x)=p_x(1)*dlog(wd)+ p_x(2)*(log(x(-1)/wd(-1))- p_x(3)*(log(ur)-  p_x(4)*log(compx(-1)))+p_x(4)*dlog(compx)+c_x(5)+p_x(6)*(t-2012)*(t<=2012)+x_ec

Cioncerning the transfers, there are two options:
· Transferring the equations one by one, according to the needs.
· Transferring all the equations at the beginning, without any change (except for replacing “c” by “p”).
We prefer the second method, which generally saves time and allows an initial check, as the result of the forecast should be the same as a simulation without any change.


producing actual forecasts
When we can decide that the model properties are acceptable (this will mean a lot more shocks than the seven in the above test) we can proceed to forecasting.
This process is much more complex than the simple “trend” simulations we have presented earlier.
As we have explained before, we can consider three or even four subperiods:
· The past year for which we know precisely some elements (like GDP at constant prices, CPI…) but in general not the whole economic equilibrium.
· The present year for which we have already some indications.
· The next two or three years for which the family of economist and economic institutions are already giving their assumptions.
· The medium-long term for which some trends already exist, and some constraints will have to be met.
These ideas can come from statisticians, experts in government policy (including civil servants working for the Ministry of Budget), expert in international issues, and the modelers themselves.
To help the process, EViews can allow reaching a target on a single variable. We propose instead a method allowing to reach a number of targets over time. Even if it does not work always, It will speed up the process in many circumstances.
How to reach a given result for some of the endogenous variables
Of course, the periods for which all the data is known will not be simulated. They are considered as history.
There are two situations in which we want the model to produce a given result for some of the endogenous.
· For the last period (or periods for a quarterly model) we already know some elements, like GDP and inflation, and we want the model to give them as an exact solution.
· We have some idea about the future evolution of some variables, to which we want the model forecast to adapt. But in this case, we are ready to accept some difference, depending on the strength of our convictions.
The two methods can use the same technology, described in detail in an attached document.
To apply it to our case, we have designed two forecasting programs, one applying the method once, presumably to the first case, the other twice, presumably for both cases in sequence.
The single period case : proj_afs2.prg
Let us explain how the first program works. Once understood, the use of the second program becomes straightforward, as well as a sequence of several applications.
To illustrate the method, we shall suppose we want the model “_mod_afs” to produce growth rates of GDP and GDP deflator equal to the “historical” value  in 2013.
The program contains a set of statements, starting with comments
 
'***************************************************************
'    User-dependent information  set 1 (start copying from here)
'
'***************************************************************
'
'     This paragraph defines the context-dependent elements

'       Now we start creating the elements required from the user

'       1 - The starting and ending periods
'       2 - The name of the model
'       3 - The two lists of variables which change status
'       4 - The target values for the new exogenous 

'        They must be typed in the appropriate spaces
'         in the lines showing : ***** User information ************

'        NO OTHER EDITING IS REQUIRED

'        THE SAME FORMAL MODEL IS USED, NO COMPILATION REQUIRED

'    We define the simulation periods

%date_1="2013"        '  ***** User information no 1 ************
%date_2="2013"        '  ***** User information no 1 ************

'       We define the sample range

delete _range
sample _range %date_1 %date_2
 

%model= "_mod_afs"      '  ***** User information  no 2 ************

'       We define the lists of elements which change status
'       exogenous to endogenous
'       endogenous to exogenous
'       The numbers must be the same

'       If not, the program stops with a message 

'        We shall set the target values (called _star) to deviations from historical values

'       We define the list of the new endogenous and exogenous (same number)

%l_endo ="gdpm  pgdpm " '  ' ***** User information no 3 ************   The list of new exogenous
%l_exo ="m_ec pq_ec " '  " '***** User information no 3 ************   The list of new endogenous

smpl _range 

'       The target values for the new exogenous

genr pgdpm_star=pgdpm   '  ***** User information no 4 ************   The target values      
genr gdpm_star=gdpm   '  ***** User information no 4 ************   The target values s  
    

smpl _range 

'*****************************************

'   End of user dependent statements (end copying here)

'*****************************************

'       We include the subroutine program

include star_subb

'       We call the subroutine (one time here, can be several with following ranges)
'       This subroutine does not have to be edited

call star  

‘      transferring the optimal values to the base 

for %1 {%l_exo}
smpl _range 
genr {%1}={%1}_0

‘      Extrapolating the assumptions

smpl 2014 2030
genr {%1}={%1}(-1)
next
The multi period case
It follows the same lines. One has just to specify different starting and ending dates. The targets can be specified once (if they are constant or have a constant growth rate) or be given different values using SMPL.
The “star_subb” program must be inserted only once ‘to avoid an error).
Connecting the applications
If the method is applied more than once, or if it is followed by a normal simulation, the exogenous must be set to the values found. This is done automatically by:

  for %1 {%l_exo}
  smpl _range 
  genr {%1}={%1}_0
  next

« l_exo » being the list of exogenous generated earlier.
If the exogenous are based on the last values, their use beyond the end of the application calls for a new computation. In our case we want the two exogenous to keep the last value computed. We will specify:

 
 smpl 2014 2030
  for %1 {%l_exo}
  genr {%1}={%1}(-1)
  next

How to chose the exogenous
There is no absolute method. However :
· They must be known to have a strong influence on the target.
In our case, the residual on household consumption has a strong impact on GDP.
· They probably should belong to the same domain. 
But this is not necessary. The residual on the wage rate should have a strong impact on exports.
· They should accept large variations (for instance the residual of an equation with a large standard error).

· In the case of several targets, they should be as uncorrelated as possible.
Once applied (and supposing the method works) the evolution of other elements will have to be checked. If it is wrong, one should move to another variable.
As to the choice of the endogenous, it is let to the modeler itself. Of course the selected variables must play an important role in the model (like GDP or CPI).
producing shocks based on forecasts
The process is exactly the same, except that to avoid the replication of the targetting process the shocks should start from the assumptions obtained at the end of the forecasting task, and saved in the final workfile. The initial computation of assumptions should be dropped.
producing stochastic simulations: 
Producing stochastic simulations under EViews is quite simple. One should simply use the following sequence:

'      s=a : we produce all results: 
'          q_b1m : the mean
'          q_b1l, q_b1h: the bounds of the 95% probability interval
'          q_b1s : the standard error

'     r=10000 : the number of replications
 
_mod_afs.solveopt(n=t,m=1000,o=g,d=d,s=a,i=p,r=10000) 
_mod_afs.append assign @all _b1
_mod_afs.exclude {%exclude}
_mod_afs.override 

'      c=f  (false): we do not consider the error from coefficients
'      p=stoc1: the page with the individual results
     
_mod_afs.stochastic(c=f,p=stoc1)     
_mod_afs.solve   

If we consider the error on coefficients, we just replace “c=f” by “c=t” (true).
We suggest the following tests:
· The evolution of the standard error relative to the mean value. For GDP, Final demand, Exports, Imorts and the value added deflator, we would use:

  show gdpm_b1s/gdmp_bm  fd_b1s/fd_bm  x_b1s/x_bm  m_b1s/m_bm  pq_b1s/pq_b1m

and display the group as a graph.
· The evolution of the confidence interval for one variable, relative to the mean, together with the mean (ratio=1) and the deterministic solution:

  show gdpm_b1l/gdpm_b1m gdpm_b1/gdpm_b1m 1 gdpm_b1hl/gdpm_b1m 


The histogram and statistics for one variable, and one period:
smpl 2020 2020
show gdpm_1
then:

View>Descriptive Statistics & Tests>Histograms & Stats



 
solving rational expectations models : 
Again, the process is quite simple. As soon as the model contains a forward variable, EViews will detect it and solutions will use an appropriate algorithm (Fair-Taylor).
No other action is needed, except for the control of the algorithm application:
· t=arg (default=“u”)     Terminal condition for forward solution: 
                “u” (user supplied- actuals), “l” (constant level), “d” (constant difference),“g” (constant growth rate).
· w=arg                             Solve direction: “t” (two-directional), “f” (forwards only).
If one wants to consider tearting several cases at the same time, a look can be cteated on the options, but the associated equation has to be reestimated inside the loop.
[bookmark: _Toc334151982][bookmark: _Toc428620230][bookmark: _Toc431790000][bookmark: _Toc431790376][bookmark: _Toc435872749][bookmark: _Toc187142636][bookmark: _Toc316680571][bookmark: _Toc318799353][bookmark: _Toc341808839]The documentation
Investment in the documentation of series produces quick returns. It can concern:

· The definition, possibly on two levels:  a short one to display titles in tables or graphs, and a long one to fully describe the concept used.
· The source: original file (and sheet), producing institution and maybe how to contact the producer.
· The units in which the series is measured
· Additional remarks, such as the quality and status (final, provisory, estimated) of each observation.
· The date of production and last update (hours and even minutes also, which can be useful to determine exactly which set of values an application has used). This information is often recorded automatically by the software.
· If pertinent, the formula used to compute it.

Example:  Wage rate = Wages paid / (employment x Number of weeks of work x weekly work duration).

EViews allows to specify the first four types, using the label command, and produces automatically the last two.

For example, a series called GDP can be defined through the sequence:


   GDP.label(c)
   GDP.label(d) Gross Domestic Product at constant prices
   GDP.label(u) In 2005 Rands 
   GDP.label(s) from the site of the World Bank
   GDP.label(r) 2013 is provisory


Which clears the contents, gives the definition, describes units, the source, and adds remarks.

· In addition, EViews 8 allows to introduce one’s own labels, for instance the country for a multinational model, the agent for an accounting one, or the fact that a series belongs to a particular model.

For instance you can use:


   HI.label(agent) Households
   MARG.label(agent) Firms


· Moreover, if the workfile window screen is in “Display+” mode, you can sort the elements according to their characteristics. In addition to the name, the type and the time of last modification (or creation) you have access to the description.

And if you right click on one of the column headings, and choose “Edit Columns” you can display additional columns for any of the label types, including the ones you have created.

This can prove quite useful, as it allows you to filter and sort on any criterion, provided you have introduced it as a label.

This criterion can be for instance:

· The agent concerned
· The country
· The association with a given model
· The formula in the model
· The formula used to create the series (if any)[footnoteRef:7] [7:  You can also use the “source”] 

· The type within this model (exogenous, endogenous, identity, behavior…)
· The sub-type: for exogenous it can be policy, foreign, structural. For endogenous it can be behavior or identity.

Once the display is produced, it can be transferred to a table, which can be edited (lines, fonts…) and used for presentations.

For instance, one can produce a table for a model, with columns for type, agent, units, source, identity / behavior…. This table can be sorted using any of the criteria.

These new functions allow table production to be integrated in the modelling process, a very powerful information tool for both model development and documentation.

For instance you could use:

 
  A_HDI.label(d) Disposable income
   U_MARG.label(d)  Margins
   A_HDI.label(model) South Africa small
   U_MARG.label(model)  USA small
   A_HDI.label(agent) Households
   U_MARG.label(agentl) Firms


and produce sorted tables according to any of the three criteria.
25

image2.emf
0

400

800

1,200

1,600

2,000

1.8e+12 1.9e+12 2.0e+12 2.1e+12 2.2e+12

Series: GDPM_B1

Sample 2013 2013

Observations 10000

Mean        2.01e+12

Median    2.01e+12

Maximum   2.23e+12

Minimum   1.77e+12

Std. Dev.    6.10e+10

Skewness    0.040597

Kurtosis    2.991337

Jarque-Bera  2.778136

Probability  0.249308



image1.emf
0

1

2

3

4

5

6

7

2020 2030 2040 2050 2060 2070 2080 2090 2100

Gowth rate of the value added deflator

Theoretical growth rate

Growth rate of value add

Theoretical growth rate

The convergence of growth rates



